



# FLR Wraps Audit Report

May 26, 2023



# Table of Contents

Summary	2
Overview	3
Issues	4
[WP-H1] Sophisticated validators can exploit the point system and rip off other honest validators	4
[WP-M2] <code>WrapDepositRedeem</code> Wrong implementation of <code>accumulatedValidatorFees()</code>	9
[WP-M3] Unable to redeem and <code>claimValidatorFees()</code> on <code>WrapDepositRedeemCustodian</code>	12
[WP-M4] Potential DoS attack by initiating a bridge transaction that cannot be fulfilled and block the entire queue	15
[WP-L5] <code>Multisig</code> Both committees cannot have the <code>maxCommitteeSize</code> (128) of members at the same time	17
[WP-L6] <code>configureValidatorFeeRecipient()</code> should be able to be called by the validator themselves	20
[WP-I7] Committee seats cannot be recycled	22
[WP-I8] Consider providing a method to revoke the votes by the signer in <code>removeSigner()</code>	24
[WP-G9] The total number of signers is known, therefore it is unnecessary to loop 256 times	26
[WP-N10] Fee-on-transfer tokens and rebasing tokens are not supported	28
Appendix	29
Disclaimer	30



## Summary

This report has been prepared for FLR Wraps smart contract, to discover issues and vulnerabilities in the source code of their Smart Contract as well as any contract dependencies that were not part of an officially recognized library. A comprehensive examination has been performed, utilizing Static Analysis and Manual Review techniques.

The auditing process pays special attention to the following considerations:

- Testing the smart contracts against both common and uncommon attack vectors.
- Assessing the codebase to ensure compliance with current best practices and industry standards.
- Ensuring contract logic meets the specifications and intentions of the client.
- Cross referencing contract structure and implementation against similar smart contracts produced by industry leaders.
- Thorough line-by-line manual review of the entire codebase by industry experts.



# Overview

## Project Summary

Project Name	FLR Wraps
Codebase	<a href="https://github.com/flrfinance/flr-wraps-contracts">https://github.com/flrfinance/flr-wraps-contracts</a>
Commit	e6564c4e47ab69137a69619413d0b4d46f749357
Language	Solidity

## Audit Summary

Delivery Date	May 26, 2023
Audit Methodology	Static Analysis, Manual Review
Total Issues	10

## [WP-H1] Sophisticated validators can exploit the point system and rip off other honest validators

High

### Issue Description

The current design of the validator points system for the distribution of validator fees is too simple, as it does not track the actual contribution in terms of the value of each execution.

As a result, the sophisticated validators can send many real bridge transactions to acquire more points and dilute the value of each point for all the other honest validators.

<https://github.com/flrfinance/flr-wraps-contracts/blob/3e26cc142f9ff23eb5370583c7fd9b66d4d4da97/src/libraries/Multisig.sol#L243-L252>

```

243 function incrementPoints(DualMultisig storage s, uint256 mask) private {
244     uint16 count = 0;
245     for (uint16 i = 0; i < maxSignersSize; i++) {
246         if ((mask & (1 << i)) != 0) {
247             s.points[i]++;
248             count++;
249         }
250     }
251     s.totalPoints += count;
252 }

```

<https://github.com/flrfinance/flr-wraps-contracts/blob/3e26cc142f9ff23eb5370583c7fd9b66d4d4da97/src/Wrap.sol#L314-L324>

```

314 function claimValidatorFees(address validator) public {
315     address feeRecipient = validatorFeeRecipients[validator];
316     uint64 totalPoints = multisig.totalPoints;
317     uint64 points = multisig.clearPoints(validator);
318     for (uint256 i = 0; i < tokens.length; i++) {
319         address token = tokens[i];
320         uint256 tokenValidatorFee = (accumulatedValidatorFees(token) *
321             points) / totalPoints;

```

```

322         IERC20(token).safeTransfer(feeRecipient, tokenValidatorFee);
323     }
324 }

```

## PoC

Given:

- validatorFeeBPS is **100** or **1%** ;
  - total committee members: **10**
  - quorum: **5**
1. Alice bridged 1M USDC and added 1k USDC of validatorFee to the **accumulatedValidatorFees** .
  2. The 5 committee members who voted each get 1 point; each point is now worth 200 USDC.
  3. The 3 sophisticated validators conspired to initiate 10 small bridge transactions, each with only 10 USDC and contributed 1 USDC for the accumulated fees in total, but acquired 30 points out of the total 50 new points.

As a result, the **totalPoints** is now **55** ; and **accumulatedValidatorFees** is now **1,001 USDC** .  
 The 3 sophisticated validators now combine for a total of 30 points, and can take out  **$1001 * 30 / 55 == 546$  USDC** .

The first 5 validators can only receive a total of **109.2 USDC** .

This is unexpected, as the first five validators would call **claimValidatorFees()** before the sophisticated validators diluted the value of their points.

## Recommendation

Consider changing to a simpler real-time settlement solution:

### Changes on **Multisig.sol**

- 1, Get rid of the points, including **incrementPoints()** ;
- 2, Add a new function called **getApprovers()** to get the list and total count of approvals;

```

255 function getApprovers(DualMultisig storage s, uint256 mask) private view returns
    (uint16[] memory, uint16) {
256     uint256 signersCount = s.firstCommitteeSize + s.secondCommitteeSize;
257     uint16[] memory approvers = new uint16[](signersCount);
258     uint16 count = 0;
259     for (uint16 i = 0; i < signersCount; i++) {
260         if ((mask & (1 << i)) != 0) {
261             approvers[count] = i;
262             count++;
263         }
264     }
265
266     return (approvers, count);
267 }

```

3, Update `tryApprove()` to return the approvers list and count.

```

276 function tryApprove(
277     DualMultisig storage s,
278     address signer,
279     bytes32 hash,
280     uint256 id
281 ) internal returns (RequestStatusTransition, uint16[] memory, uint16) {
282     Request storage request = s.requests[hash];
283     // If the request has already been accepted
284     // then simply return.
285     if (request.status == RequestStatus.Accepted) {
286         return (RequestStatusTransition.Unchanged, new uint16[](0), 0);
287     }
288
289     @@ 289,299 @@
290
300
301     uint256 signerMask = 1 << signerInfo.index;
302     // Check if the signer has already signed.
303     if ((signerMask & request.approvers) != 0) {
304         return (RequestStatusTransition.Unchanged, new uint16[](0), 0);
305     }
306

```

```

@@ 307,313 @@
314
315     // If quorum has been reached, increment the number of points.
316     if (
317         request.approvalsFirstCommittee >=
318         s.firstCommitteeAcceptanceQuorum &&
319         request.approvalsSecondCommittee >=
320         s.secondCommitteeAcceptanceQuorum
321     ) {
322         request.status = RequestStatus.Accepted;
323         s.approvedRequests[id] = hash;
324
325         (uint16[] memory approvers, uint16 count) = getApprovers(s,
request.approvers);
326         return (RequestStatusTransition.UndecidedToAccepted, approvers, count);
327     } else if (request.status == RequestStatus.NULL) {
328         // If this is the first approval, change the request status
329         // to undecided.
330         request.status = RequestStatus.Undecided;
331         return (RequestStatusTransition.NULLToUndecided, new uint16[](0), 0);
332     }
333     return (RequestStatusTransition.Unchanged, new uint16[](0), 0);
334 }

```

## Changes on `wrap.sol`

1, Add a new mapping for the fee balance of each token for each validator;

```

39     mapping(address => mapping(uint256 => uint256)) public feeBalance;

```

2, Change `approveExecute()` to add fee to validators' balance after execution;

```

166     function approveExecute(
167         uint256 id,
168         address mirrorToken,
169         uint256 amount,
170         address to
171     ) public isNotPaused isValidMirrorTokenAmount(mirrorToken, amount) {
172         // If the request ID is lower than the last executed ID then simply ignore the
request.

```



```

173     if (id < multisig.nextExecutionIndex) {
174         return;
175     }
176
177     bytes32 hash = hashRequest(id, mirrorToken, amount, to);
178     (Multisig.RequestStatusTransition transition, uint16[] approvers, uint16
179     approverCount) = multisig.tryApprove(
180         msg.sender,
181         hash,
182         id
183     );
184     if (transition == Multisig.RequestStatusTransition.NULLToUndecided) {
185         emit Requested(id, mirrorToken, amount, to);
186     }
187
188     if (multisig.tryExecute(hash, id)) {
189         address token = mirrorTokens[mirrorToken]; // token -> address()
190         uint256 fee = onExecute(token, amount, to);
191
192         uint256 singerFee = fee / approverCount;
193         for(uint16 i; i < approverCount; i++){
194             feeBalance[token][approvers[i]] += singerFee;
195         }
196         emit Executed(id, token, amount - fee, to, fee);
197     }

```

3, Change the `claimValidatorFees()` function to use `feeBalance` .

## Status

✓ Fixed

## [WP-M2] WrapDepositRedeem Wrong implementation of accumulatedValidatorFees()

Medium

### Issue Description

<https://github.com/flrfinance/flr-wraps-contracts/blob/89eca4b09a4a08857e96bd7f49b4941f31863386/src/WrapDepositRedeem.sol#L24-L32>

```

24  function accumulatedValidatorFees(address token)
25      public
26      view
27      virtual
28      override(IWrap, Wrap)
29      returns (uint256)
30  {
31      return IERC20(token).balanceOf(address(this));
32  }
```

`accumulatedValidatorFees()` should not return all the account balance directly.

<https://github.com/flrfinance/flr-wraps-contracts/blob/89eca4b09a4a08857e96bd7f49b4941f31863386/src/WrapDepositRedeem.sol#L51-L58>

```

51  function onExecute(
52      address token,
53      uint256 amount,
54      address to
55  ) internal virtual override returns (uint256 fee) {
56      fee = calculateFee(amount, validatorFeeBPS);
57      IERC20(token).safeTransfer(to, amount - fee);
58  }
```

According to `onExecute()`, only part of the balance is the accumulated validator fees;

The majority of the balance should be the reserve for future redemption.

## Recommendation

Consider changing to:

```

14  contract WrapDepositRedeem is IWrapDepositRedeem, Wrap {
    @@ 15,17 @@
18
19      mapping(address => uint256) public accumulatedValidatorFeeAmounts;
20
    @@ 21,24 @@
25
26      /// @inheritdoc IWrap
27      function accumulatedValidatorFees(
28          address token
29      ) public view virtual override(IWrap, Wrap) returns (uint256) {
30          return accumulatedValidatorFeeAmounts[token];
31      }
32
    @@ 33,45 @@
46
47      /// @inheritdoc Wrap
48      function onExecute(
49          address token,
50          uint256 amount,
51          address to
52      ) internal virtual override returns (uint256 fee) {
53          fee = calculateFee(amount, validatorFeeBPS);
54          accumulatedValidatorFeeAmounts[token] += fee;
55          IERC20(token).safeTransfer(to, amount - fee);
56      }
    @@ 57,65 @@
66
67      function claimValidatorFees(address validator) public override {
68          address feeRecipient = validatorFeeRecipients[validator];
69          uint64 totalPoints = multisig.totalPoints;
70          uint64 points = multisig.clearPoints(validator);
71          for (uint256 i = 0; i < tokens.length; i++) {
72              address token = tokens[i];
73              uint256 tokenValidatorFee = (accumulatedValidatorFees(token) *
74                  points) / totalPoints;

```

```
75     accumulatedValidatorFeeAmounts[token] -= tokenValidatorFee;
76     IERC20(token).safeTransfer(feeRecipient, tokenValidatorFee);
77 }
78 }
79 }
```

## Status

✓ Fixed

## [WP-M3] Unable to redeem and `claimValidatorFees()` on `WrapDepositRedeemCustodian`

Medium

### Issue Description

`WrapDepositRedeemCustodian` inherit `WrapDepositRedeem` 's `accumulatedValidatorFees()` :

<https://github.com/flrfinance/flr-wraps-contracts/blob/89eca4b09a4a08857e96bd7f49b4941f31863386/src/WrapDepositRedeem.sol#L24-L32>

```

24  function accumulatedValidatorFees(address token)
25      public
26      view
27      virtual
28      override(IWrap, Wrap)
29      returns (uint256)
30  {
31      return IERC20(token).balanceOf(address(this));
32  }

```

However, since all the deposits will be transferred to the `custodian` address, including the `fee`, there will always be a balance of 0.

<https://github.com/flrfinance/flr-wraps-contracts/blob/3e26cc142f9ff23eb5370583c7fd9b66d4d4da97/src/WrapDepositRedeemCustodian.sol#L28-L34>

```

28  function onDeposit(
29      address token,
30      uint256 amount
31  ) internal override returns (uint256) {
32      IERC20(token).safeTransferFrom(msg.sender, custodian, amount);
33      return depositFees(amount);
34  }

```

As a result, the validators will not be able to `claimValidatorFees()` .

Furthermore, the users also cannot redeem as there is no transfer in the `onExecute()` function:

[https://github.com/flrfinance/flr-wraps-contracts/blob/](https://github.com/flrfinance/flr-wraps-contracts/blob/3e26cc142f9ff23eb5370583c7fd9b66d4d4da97/src/WrapDepositRedeemCustodian.sol#L37-L43)

[3e26cc142f9ff23eb5370583c7fd9b66d4d4da97/src/WrapDepositRedeemCustodian.sol#L37-L43](https://github.com/flrfinance/flr-wraps-contracts/blob/3e26cc142f9ff23eb5370583c7fd9b66d4d4da97/src/WrapDepositRedeemCustodian.sol#L37-L43)

```

37  function onExecute(
38      address,
39      uint256 amount,
40      address
41  ) internal view override returns (uint256 fee) {
42      fee = calculateFee(amount, validatorFeeBPS);
43  }

```

## Recommendation

Consider changing to:

```

12  contract WrapDepositRedeemCustodian is WrapDepositRedeem {
13      @@ 13,25 @@
26
27      function accumulatedValidatorFees(
28          address token
29      ) public view override returns (uint256) {
30          return IERC20(token).balanceOf(address(this));
31      }
32
33      @@ 33,40 @@
41
42      /// @inheritdoc WrapDepositRedeem
43      function onExecute(
44          address,
45          uint256 amount,
46          address
47      ) internal view override returns (uint256 fee) {
48          fee = calculateFee(amount, validatorFeeBPS);
49          IERC20(token).transferFrom(custodian, to, amount - fee);
50          IERC20(token).transferFrom(custodian, address(this), fee);
51      }
52  }

```



## Status

✓ Fixed

## [WP-M4] Potential DoS attack by initiating a bridge transaction that cannot be fulfilled and block the entire queue

Medium

### Issue Description

<https://github.com/flrfinance/flr-wraps-contracts/blob/3e26cc142f9ff23eb5370583c7fd9b66d4d4da97/src/libraries/Multisig.sol#L327-L337>

```
327 function tryExecute(  
328     DualMultisig storage s,  
329     bytes32 hash,  
330     uint256 id  
331 ) internal returns (bool) {  
332     if (id == s.nextExecutionIndex && s.approvedRequests[id] == hash) {  
333         s.nextExecutionIndex++;  
334         return true;  
335     }  
336     return false;  
337 }
```

In the current implementation, requests must be executed in order and cannot be skipped. Otherwise, subsequent requests will be blocked.

<https://github.com/flrfinance/flr-wraps-contracts/blob/3e26cc142f9ff23eb5370583c7fd9b66d4d4da97/src/WrapDepositRedeem.sol#L46-L53>

```
46 function onExecute(  
47     address token,  
48     uint256 amount,  
49     address to  
50 ) internal virtual override returns (uint256 fee) {  
51     fee = calculateFee(amount, validatorFeeBPS);  
52     IERC20(token).safeTransfer(to, amount - fee);  
53 }
```



However, `onExecute()` depends on the successful call to the external contract ( `token` ).

For example, if the recipient is blacklisted in USDC, the whole bridge will be blocked.

Also, one can intentionally bridge some USDC to a blacklisted address in order to initiate a DoS attack on the system.

## Recommendation

Consider changing `IERC20(token).safeTransfer(to, amount - fee)` to require `to` to collect it themselves (from PUSH to PULL), or remove the sequential execution limitation, or allow the admin to bump `nextExecutionIndex` .

## Status

✓ Fixed

## [WP-L5] Multisig Both committees cannot have the maxCommitteeSize (128) of members at the same time

Low

### Issue Description

Per the comments:

a multisig with two committees ... Each committee cannot have more than 128 members

In `Multisig.sol: L81-83, L89`, `approvers` is a bitmask typed `uint256`, representing a bijection between bits and signer indices.

More specifically, `approver.bitAt(i) ↔ indexi`, where `index` is defined at L97.

As `index` is declared as `uint8`, its value ranges from 0 to 255.

However, the index is actually assigned values from 1 to 256:

<https://github.com/flrfinance/flr-wraps-contracts/blob/3e26cc142f9ff23eb5370583c7fd9b66d4d4da97/src/libraries/Multisig.sol#L199-L225>

```

199  function addSigner(
200      DualMultisig storage s,
201      address signer,
202      bool isFirstCommittee
203  ) internal {
204      uint8 committeeSize = (
205          isFirstCommittee ? s.firstCommitteeSize : s.secondCommitteeSize
206      );
207      if (committeeSize == maxCommitteeSize) {
208          revert MaxCommitteeSizeReached();
209      }
210
211      SignerInfo storage signerInfo = s.signers[signer];
212      if (signerInfo.status != SignerStatus.Uninitialized) {
213          revert SignerAlreadyExists(signer);
214      }
215  }

```

```

216     if (isFirstCommittee) {
217         s.firstCommitteeSize++;
218         signerInfo.status = SignerStatus.FirstCommittee;
219     } else {
220         s.secondCommitteeSize++;
221         signerInfo.status = SignerStatus.SecondCommittee;
222     }
223
224     signerInfo.index = s.firstCommitteeSize + s.secondCommitteeSize;
225 }

```

That is, when one of the committees already have the `maxCommitteeSize` (128) of members, then the other committee can not have the `maxCommitteeSize` (128) number of members, as when adding the 128th member, `signerInfo.index` will overflow.

## Recommendation

Consider changing to:

```

265 function addSigner(
266     DualMultisig storage s,
267     address signer,
268     bool isFirstCommittee
269 ) internal {
270     uint8 committeeSize = (
271         isFirstCommittee ? s.firstCommitteeSize : s.secondCommitteeSize
272     );
273     if (committeeSize == maxCommitteeSize) {
274         revert MaxCommitteeSizeReached();
275     }
276
277     SignerInfo storage signerInfo = s.signers[signer];
278     if (signerInfo.status != SignerStatus.Uninitialized) {
279         revert SignerAlreadyExists(signer);
280     }
281
282     signerInfo.index = s.firstCommitteeSize + s.secondCommitteeSize;
283
284     if (isFirstCommittee) {
285         s.firstCommitteeSize++;

```

```
286         signerInfo.status = SignerStatus.FirstCommittee;
287     } else {
288         s.secondCommitteeSize++;
289         signerInfo.status = SignerStatus.SecondCommittee;
290     }
291 }
```

So that the `signerInfo.index` starts with `0` instead `1` .

## Status

 Acknowledged

## [WP-L6] `configureValidatorFeeRecipient()` should be able to be called by the validator themselves

Low

### Issue Description

<https://github.com/flrfinance/flr-wraps-contracts/blob/3e26cc142f9ff23eb5370583c7fd9b66d4d4da97/src/Wrap.sol#L306-L311>

```
306 function configureValidatorFeeRecipient(  
307     address validator,  
308     address feeRecipient  
309 ) external onlyRole(DEFAULT_ADMIN_ROLE) {  
310     validatorFeeRecipients[validator] = feeRecipient;  
311 }
```

The `feeRecipient` should be configured by the validator themselves, rather than the admin.

Conventionally, if a specific recipient address is not set, it should fall back to the original validator address.

### Recommendation

Consider changing to:

```
306 function configureValidatorFeeRecipient(  
307     address validator,  
308     address feeRecipient  
309 ) external {  
310     if (msg.sender != validator) revert Auth();  
311     validatorFeeRecipients[validator] = feeRecipient;  
312 }
```

Also, consider disallowing setting the `feeRecipient` as `address(0)` to avoid potential misapplication.



Or, consider supporting fallingback to `validator` address when the `feeRecipient` is set as `address(0)` .

## Status

 Acknowledged

## [WP-I7] Committee seats cannot be recycled

### Informational

### Issue Description

The total number of both committees combined is 256:

<https://github.com/flrfinance/flr-wraps-contracts/blob/3e26cc142f9ff23eb5370583c7fd9b66d4d4da97/src/libraries/Multisig.sol#L73-L90>

```

73      /// @dev Request info.
74      /// @param approvalsFirstCommittee Number of approvals
75      /// by the first committee.
76      /// @param approvalsSecondCommittee Number of approvals
77      /// by the second committee.
78      /// @param status Status of the request.
79      /// @param approvers Bitmask for signers from the two
80      /// committees who have accepted the request.
81      /// @notice Approvers is a bitmask. For example, a set bit at
82      /// position 2 in the approvers bitmask indicates that the
83      /// signer with index 2 has approved the request.
84      struct Request {
85          uint8 approvalsFirstCommittee; // slot 1 (0 - 7 bits)
86          uint8 approvalsSecondCommittee; // slot 1 (8 - 15 bits)
87          RequestStatus status; // slot 1 (16 - 23 bits)
88          // slot 1 (23 - 255 spare bits)
89          uint256 approvers; // slot 2
90      }

```

Every new signer will take a new seat ( `index` ) instead of reusing a previously removed signer's seat ( `removeSigner()` ):

<https://github.com/flrfinance/flr-wraps-contracts/blob/3e26cc142f9ff23eb5370583c7fd9b66d4d4da97/src/libraries/Multisig.sol#L199-L225>

```

199  function addSigner(
200      DualMultisig storage s,
201      address signer,
202      bool isFirstCommittee

```

```


203 ) internal {
204     uint8 committeeSize = (
205         isFirstCommittee ? s.firstCommitteeSize : s.secondCommitteeSize
206     );
207     if (committeeSize == maxCommitteeSize) {
208         revert MaxCommitteeSizeReached();
209     }
210
211     SignerInfo storage signerInfo = s.signers[signer];
212     if (signerInfo.status != SignerStatus.Uninitialized) {
213         revert SignerAlreadyExists(signer);
214     }
215
216     if (isFirstCommittee) {
217         s.firstCommitteeSize++;
218         signerInfo.status = SignerStatus.FirstCommittee;
219     } else {
220         s.secondCommitteeSize++;
221         signerInfo.status = SignerStatus.SecondCommittee;
222     }
223
224     signerInfo.index = s.firstCommitteeSize + s.secondCommitteeSize;
225 }

```

Over a sufficient period of time, there will be more and more seats that become unusable, and the valid committee members will get fewer.

By the time that the total number of valid seats is lower than the quorum, the system will be forced to deprecate the current set of contracts and deploy new ones, or lower the quorum which will also lower the security level at the same time.

## Status

 Acknowledged



## [WP-I8] Consider providing a method to revoke the votes by the signer in `removeSigner()`

### Informational

### Issue Description

<https://github.com/flrfinance/flr-wraps-contracts/blob/3e26cc142f9ff23eb5370583c7fd9b66d4d4da97/src/Wrap.sol#L298-L303>

```

298  function removeValidator(
299      address validator
300  ) external onlyRole(DEFAULT_ADMIN_ROLE) {
301      claimValidatorFees(validator);
302      multisig.removeSigner(validator);
303  }

```

The current implementation of `removeValidator()` will trigger `claimValidatorFees()` and change the `status` of the `signer` to `Removed`.

However, the votes cast by the signer on the pending executions will remain valid.

This can be a problem if the number of signers got compromised (so that needed to be removed) is near the quorum.

### Recommendation

Consider introducing a new method that allows the admin to also decrease the votes:

```

298  function removeValidator(
299      address signer,
300      bytes32[] memory hashes,
301  ) external onlyRole(DEFAULT_ADMIN_ROLE) {
302      claimValidatorFees(signer);
303      SignerInfo memory signerInfo = s.signers[signer];
304      // decode and loop hashes and remove the vote from each one
305      for (uint256 i = 0; i < hashes.length; ++i) {
306          Request storage request = s.requests[hash];

```

```
307     if (request.status != RequestStatus.Undecided) continue;
308     uint256 signerMask = 1 << signerInfo.index;
309     // make sure the signer has already signed.
310     if ((signerMask & request.approvers) != 0) {
311         request.approvers &= ~signerMask;
312         if (signerInfo.status == SignerStatus.FirstCommittee) {
313             --request.approvalsFirstCommittee;
314         } else {
315             --request.approvalsSecondCommittee;
316         }
317     }
318 }
319 multisig.removeSigner(signer);
320 }
```

## Status

 Acknowledged

## [WP-G9] The total number of signers is known, therefore it is unnecessary to loop 256 times

Gas

### Issue Description

<https://github.com/flrfinance/flr-wraps-contracts/blob/89eca4b09a4a08857e96bd7f49b4941f31863386/src/libraries/Multisig.sol#L246-L255>

```

246  function incrementPoints(DualMultisig storage s, uint256 mask) private {
247      uint16 count = 0;
248      for (uint16 i = 0; i < maxSignersSize; i++) {
249          if ((mask & (1 << i)) != 0) {
250              s.points[i]++;
251              count++;
252          }
253      }
254      s.totalPoints += count;
255  }

```

### Recommendation

```

246  function incrementPoints(DualMultisig storage s, uint256 mask) private {
247      uint16 count = 0;
248      uint256 signersCount = s.firstCommitteeSize + s.secondCommitteeSize;
249      for (uint16 i = 1; i <= signersCount; i++) {
250          if ((mask & (1 << i)) != 0) {
251              s.points[i]++;
252              count++;
253          }
254      }
255      s.totalPoints += count;
256  }

```



## Status

✓ Fixed

## [WP-N10] Fee-on-transfer tokens and rebasing tokens are not supported

### Issue Description

It is worth noting that the system is not designed to support fee-on-transfer tokens and rebasing tokens.

The admin should not add these tokens.

### Status

 Acknowledged



# Appendix

## Timeliness of content

The content contained in the report is current as of the date appearing on the report and is subject to change without notice, unless indicated otherwise by WatchPug; however, WatchPug does not guarantee or warrant the accuracy, timeliness, or completeness of any report you access using the internet or other means, and assumes no obligation to update any information following publication.

## Disclaimer

This report is based on the scope of materials and documentation provided for a limited review at the time provided. Results may not be complete nor inclusive of all vulnerabilities. The review and this report are provided on an as-is, where-is, and as-available basis. You agree that your access and/or use, including but not limited to any associated services, products, protocols, platforms, content, and materials, will be at your sole risk. Smart Contract technology remains under development and is subject to unknown risks and flaws. The review does not extend to the compiler layer, or any other areas beyond the programming language, or other programming aspects that could present security risks. A report does not indicate the endorsement of any particular project or team, nor guarantee its security. No third party should rely on the reports in any way, including for the purpose of making any decisions to buy or sell a product, service or any other asset. To the fullest extent permitted by law, we disclaim all warranties, expressed or implied, in connection with this report, its content, and the related services and products and your use thereof, including, without limitation, the implied warranties of merchantability, fitness for a particular purpose, and non-infringement. We do not warrant, endorse, guarantee, or assume responsibility for any product or service advertised or offered by a third party through the product, any open source or third-party software, code, libraries, materials, or information linked to, called by, referenced by or accessible through the report, its content, and the related services and products, any hyperlinked websites, any websites or mobile applications appearing on any advertising, and we will not be a party to or in any way be responsible for monitoring any transaction between you and any third-party providers of products or services. As with the purchase or use of a product or service through any medium or in any environment, you should use your best judgment and exercise caution where appropriate. FOR AVOIDANCE OF DOUBT, THE REPORT, ITS CONTENT, ACCESS, AND/OR USAGE THEREOF, INCLUDING ANY ASSOCIATED SERVICES OR MATERIALS, SHALL NOT BE CONSIDERED OR RELIED UPON AS ANY FORM OF FINANCIAL, INVESTMENT, TAX, LEGAL, REGULATORY, OR OTHER ADVICE.